Диференциално и интегрално смятане със СКА Mathematica

Граница на редица и граница на функция. Лява и дясна граница.

$$\texttt{Limit}\left[\left(1+\frac{1}{n}\right)^{n},\ n \to \texttt{Infinity}\right]$$

e

$$\texttt{Limit}\left[\frac{\texttt{Sin}[x]}{x},\ x \to 0\right]$$

1

$$h[x\_] := \frac{1}{1+2^{-1/x}}$$

```
Plot[h[x], {x, -10, 10}, PlotRange → All]
```



Ако дадена граница не съществува, резултатът от функцията **Limit** ще бъде ***Indeterminate***. За да пресметнем лява и дясна граница е необходимо да зададем опцията ***Direction***, със съответни стойности *“FromBelow”* и *“FromAbove”*.

```
Limit[h[x], x → 0] (*По-новите версии на системата,
ще върнат като резултат Indeterminate, в случай че границата не съществува*)
```

1

```
Limit[h[x], x → 0 , Direction → 1] (*В по-новите версии,
опциите на Direction са "FromAbove" и "FromBelow"*)
```

0

```
Limit[h[x], x → 0 , Direction → -1]
```

1

```
Sum[n/(n^4 + n^2 + 1), {n, 1, Infinity}] (*Сума на ред*)
```

$\frac{1}{2}$

```
Sum[1/Factorial[n] (n/E)^n, {n, 1, Infinity}] (*Редът не е сходящ*)
```

Sum: Sum does not converge.

$\sum_{n=1}^{\infty} \frac{e^{-n} n^n}{n!}$

```
Product[i^2, {i, 1, n}]
```

$(n!)^2$

```
Product[i^2, {i, 1, 10}]
```

13 168 189 440 000

Производна на функция

```
f[x_] := Cos[x] x^6
```

```
f'[x]
```

$6 x^5 \operatorname{Cos}[x] - x^6 \operatorname{Sin}[x]$

```
f''[x]
```

$30 x^4 \operatorname{Cos}[x] - x^6 \operatorname{Cos}[x] - 12 x^5 \operatorname{Sin}[x]$

```
D[f[x], {x, 4}]
```

$360 x^2 \operatorname{Cos}[x] - 180 x^4 \operatorname{Cos}[x] + x^6 \operatorname{Cos}[x] - 480 x^3 \operatorname{Sin}[x] + 24 x^5 \operatorname{Sin}[x]$

```
D[f[x], {x, 4}] /. x → 0.5 (*стойност в точка*)
```

40.7174

Развитие в ред на Тейлър

```
Series[E^x Cos[2 x], {x, 0, 5}]
```

$1 + x - \frac{3 x^2}{2} - \frac{11 x^3}{6} - \frac{7 x^4}{24} + \frac{41 x^5}{120} + O[x]^6$

```
Series[E^x Cos[2 x], {x, 0, 5}] // Normal
  (*Normal извежда полинома до съответната степен без остатъчния член*)
```

$1 + x - \frac{3 x^2}{2} - \frac{11 x^3}{6} - \frac{7 x^4}{24} + \frac{41 x^5}{120}$

Екстремуми на функции

Глобални екстремуми

Minimize и **Maximize** търсят глобален екстремум в инервала (-∞,∞), освен ако допълнителни ограничения за областта на търсене не са зададени. Първият компонент на резултата е екстремалната стойност на функцията, а вторият - аргументът, при който се достига тя. **Minimize** и **Maximize** могат да работят с функции на произволен брой независими променливи.

```
Minimize[2 x^2 - 3 x + 5, x]
```

$\left\{\frac{31}{8}, \left\{x \to \frac{3}{4}\right\}\right\}$

```
Minimize[(x y - 3)^2 + 1, {x, y}]
```

```
{1, {x → -1, y → -3}}
```

В случай че дадена функция не е ограничена, ще получим предупреждение.

```
{Minimize[x + 2 Sin[x], x], Maximize[x + 2 Sin[x], x]}
```

Minimize: The minimum is not attained at any point satisfying the given constraints.

Maximize: The maximum is not attained at any point satisfying the given constraints.

```
{{-∞, {x → -∞}}, {∞, {x → ∞}}}
```

```
Plot[x + 2 Sin[x], {x, -10, 10}]
```



Можем да зададем допълнителни ограничения за областта на търсене **във вид на неравенства.**

```
Maximize[x + 2 Sin[x], -10 ≤ x ≤ 10, x] (*Пресмята симоволно*)
```

$\left\{\sqrt{3} + \frac{8\pi}{3}, \left\{x \to \frac{8\pi}{3}\right\}\right\}$

NMinimize и **NMaximize** са съответните функции за числено намиране на глобални екстремуми.

```
NMaximize[x + 2 Sin[x], -10 ≤ x ≤ 10, x] (*Пресмята числено*)
```

```
{10.1096, {x → 8.37758}}
```

```
NMinimize[x + 2 Sin[x], -10 ≤ x ≤ 10, x]
```

```
{-10.1096, {x → -8.37758}}
```

Локални екстремуми

FindMaximum/FindMinimum търсят **локален екстремум** на дадена функкция, започвайки от подадена от потребителя стойност, като също могат да приемат ограничения за областта на търсене във вид на неравенства.

```
Plot[x + 2 Sin[x], {x, -10, -1}]
```




```
FindMaximum[x + 2 Sin[x], {x, -5}]
```

{-2.45674, {x → -4.18879}}

```
FindMaximum[x + 2 Sin[x], {x, -8}]
```

FindMaximum: The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient increase in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances.

{85.5079, {x → 83.7758}}

```
FindMaximum[{x + 2 Sin[x], -10 ≤ x ≤ -4}, {x, -8}]
```

{-2.45674, {x → -4.18879}}

Интегриране

Символно и числено интегриране

Фунцкията **Integrate** може да се използва за аналитично премсятане на неопределени и определни интеграли. Голяма част от инегралите, които се налага да се решават на практика, обаче, **не могат да бъдат решени аналитично**. В такъв случай, можем да получим числена апроксимация на стойностие им с помощта на функцията **NIntegrate.**

Integrate$\left[\frac{1}{x^3+1}, x\right]$ (*Неопределен интеграл*)

$$\frac{\text{ArcTan}\left[\frac{-1+2x}{\sqrt{3}}\right]}{\sqrt{3}} + \frac{1}{3}\text{Log}[1+x] - \frac{1}{6}\text{Log}\left[1-x+x^2\right]$$

Integrate$\left[\frac{1}{x^3+1}, \{x, 0, 1\}\right]$ (*Определен инреграл*)

$$\frac{1}{18}\left(2\sqrt{3}\,\pi + \text{Log}[64]\right)$$

NIntegrate$\left[\frac{1}{x^3+1}, \{x, 0, 1\}\right]$ (*За определени интеграли - дава числена апроксимация*)

0.835649

```
Integrate[Sin[x]/Log[x], {x, 2, 5}] (*Този интеграл,
например, не може да бъде решен аналитично*)
```

$$\int_2^5 \frac{\mathrm{Sin}[x]}{\mathrm{Log}[x]} \, dx$$

```
NIntegrate[Sin[x]/Log[x], {x, 2, 5}] (*Но може да бъде решен числено*)
```

-0.20523

Многократни интеграли

За интегриране върху правоъгълна област, можем просто да изброим границите на интегриране на всяка от променливите:

```
Integrate[Log[y]/(x y), {x, 1, 5}, {y, 1, 3}]
```

$\frac{1}{2} \mathrm{Log}[3]^2 \mathrm{Log}[5]$

Можем да интегрираме и върху произволна област.
Нека да разгледаме задачата за намиране на обема на тялото, образувано от пресичането на равнините $z_1 = 8 - 3x - 2y$ и $z_2 = 3x + y - 4$ в първи октант.



Можем да определим границите на интегриране, като използваме факта, че разглеждаме равнините само в първи октант и открием правата, в която те се пресичат:

```
Reduce[{8 - 3 x - 2 y == 3 x + y - 4, x ≥ 0, y ≥ 0}, {x, y}]
```

0 ≤ x ≤ 2 && y == 4 - 2 x

```
Integrate[8 - 3 x - 2 y - (3 x + y - 4), {x, 0, 2}, {y, 0, 4 - 2 x}]
```

16

Друга възможност за пресмятане на интеграла е да дефинираме геометричната област **R** и да я използваме директно във функцията **Integrate.**

```
R = ImplicitRegion[{0 ≤ x ≤ 2, 0 ≤ y ≤ 4 - 2 x}, {x, y}];
Integrate[8 - 3 x - 2 y - (3 x + y - 4), {x, y} ∈ R] (*Символът "∈",
можем да получим като използваме клавишната комбинация Esc +
 "el" + Esc - това е съкратен запис на командата Element[{x,y},R]*)
```

16

И разбира се, има още една възможност - да използваме троен интеграл:

```
Integrate[1, {x, 0, 2}, {y, 0, 4 - 2 x}, {z, 3 x + y - 4, 8 - 3 x - 2 y}] // Simplify
R3D = ImplicitRegion[{0 ≤ x ≤ 2, 0 ≤ y ≤ 4 - 2 x, 3 x + y - 4 ≤ z ≤ 8 - 3 x - 2 y}, {x, y, z}];
Integrate[1, {x, y, z} ∈ R3D] // Simplify
```

16

16